Hyperbolisches Paraboloid

Prof. Dr. Dörte Haftendorn, MuPAD 4,   Juni 08 Update 29. Juni 08
Web: http://haftendorn.uni-lueneburg.de   www.mathematik-verstehen.de

++

```
plotfunc3d(x*y)
```




```
ger:=plot::Curve3d([-t,a,-a*t+25],t=-5..5,a=-5..5,
    LineWidth=1,LineColor=[0,1,0]);
hypar:=plot::Function3d(x*y+25,x=-5..5,y=-5..5);
```

$\text{plot::Curve3d}([-t,\ a,\ -a\cdot t+25],\ t=-5..5)$

$\text{plot::Function3d}(x\cdot y+25,\ x=-5..5,\ y=-5..5)$

```
plot(ger,hypar)
```



animieren durch Anklicken!

Das hyperbolische Paraboloid wird durch eine "geschraubte" Gerade erzeugt.

erzeugt.
##
Volumen unter diesem Dach bis zur Nullebene

```
int(int(x*y+25, x=-5..5), y=-5..5)
```

$$2500$$

Das ist genau der habe Kasten.
##
Ansehen der Schnitte mit waagerechten Ebenen.

```
eb:=plot::Function3d(b,x=-5..5,y=-5..5,b=0..50,FillColor
=[0,1,0], FillColor2=[1,1,0] );
```

$$\text{plot::Function3d}(b,\ x = -5..5,\ y = -5..5)$$

```
plot(hypar,eb)
```



animieren durch Anklicken!

Diese Schnitte sind Hyperbeln.
###
Drehen um PI/4

```
A:=matrix([[cos(t),-sin(t),0],[sin(t),cos(t),0],
[0,0,1]])|t=PI/4;
```

$$\begin{pmatrix} \frac{\sqrt{2}}{2} & -\frac{\sqrt{2}}{2} & 0 \\ \frac{\sqrt{2}}{2} & \frac{\sqrt{2}}{2} & 0 \\ 0 & 0 & 1 \end{pmatrix}$$

```
hypparvek:=matrix([t,r,t*r])
```

$$\begin{pmatrix} t \\ r \\ r \cdot t \end{pmatrix}$$


```
hyd:=A*hypparvek
```

$$\begin{pmatrix} \frac{\sqrt{2}\cdot t}{2} - \frac{\sqrt{2}\cdot r}{2} \\ \frac{\sqrt{2}\cdot r}{2} + \frac{\sqrt{2}\cdot t}{2} \\ \ldots \end{pmatrix}$$

$$\begin{pmatrix} \frac{\sqrt{2}\cdot t}{2} - \frac{\sqrt{2}\cdot r}{2} \\ \frac{\sqrt{2}\cdot r}{2} + \frac{\sqrt{2}\cdot t}{2} \\ r\cdot t \end{pmatrix}$$

```
X:=hyd[1]; Y:=hyd[2]; Z:=hyd[3]
```

$$\frac{\sqrt{2}\cdot t}{2} - \frac{\sqrt{2}\cdot r}{2}$$

$$\frac{\sqrt{2}\cdot r}{2} + \frac{\sqrt{2}\cdot t}{2}$$

$$r\cdot t$$

```
(X+Y)*(X-Y)
```

$$-2\cdot r\cdot t$$

Also gilt

```
hydreh:=x^2-y^2=2*z;
```

$$x^2 - y^2 = 2\cdot z$$

Noch um 25 nach oben schieben

```
hyparim:=plot::Implicit3d(x^2-y^2+2*(z-25)=0,x=-5..5,
y=-5..5,z=0..50)
```

$$\text{plot::Implicit3d}(2\cdot z + x^2 - y^2 - 50,\ x = -5..5,\ y = -5..5,\ z = 0..50)$$

```
plot(hyparim,eb)
```



animieren durch Anklicken! 3

Auch die wandernde Grade kann man drehen um 45°

```
gerdre:=A*matrix([-t,a,-a*t+25])
```

$$\begin{pmatrix} -\frac{\sqrt{2}\cdot a}{2} - \frac{\sqrt{2}\cdot t}{2} \\ \frac{\sqrt{2}\cdot a}{2} - \frac{\sqrt{2}\cdot t}{2} \\ \ldots \end{pmatrix}$$

$$\begin{pmatrix} -\frac{\sqrt{2}\cdot a}{2}-\frac{\sqrt{2}\cdot t}{2} \\ \frac{\sqrt{2}\cdot a}{2}-\frac{\sqrt{2}\cdot t}{2} \\ 25-a\cdot t \end{pmatrix}$$

```
gerdr:=plot::Curve3d(gerdre,t=-5..5,a=-5..5,
    LineWidth=1,LineColor=[0,1,0]);
```

$$\text{plot::Curve3d}\left(\left[-\frac{\sqrt{2}\cdot a}{2}-\frac{\sqrt{2}\cdot t}{2}, \frac{\sqrt{2}\cdot a}{2}-\frac{\sqrt{2}\cdot t}{2}, -a\cdot t+25\right], t=-5..5\right)$$

```
plot(hyparim,gerdr)
```



animieren durch Anklicken!

Nun sind die Hauptschnitte also Parabeln, die Schnitte mit horizontalen Ebenen sind immernoch Hyperbeln. darum also der Name
hyperbolisches Paraboloid
Natürlich liegt die gedrehte Gerade auch darauf.

Flächen, die durch bewegte Geraden entstehen heißen Regelflächen

Sie lassen sich leicht in Beton bauen.
Das hyperbolische Paraboloid ist also eine Regelfläche,
der Kegel, auch der Kegelstumpf ist also eine Regelfläche,
die Schraubenfläche ist also eine Regelfläche,
das einschalige Hyperboloid ist also eine Regelfläche,